

HEPHY-PUB 690/98
UWThPh-1998-29
June 1998

BOUNDS ON SCHRÖDINGER ENERGIES

An Illustration of the Application of Mathematica 3.0

HAN Liang

Department of Modern Physics,
University of Science and Technology of China,
Hefei, China

Wolfgang LUCHA

Institut für Hochenergiephysik,
Österreichische Akademie der Wissenschaften,
Nikolsdorfergasse 18, A-1050 Wien, Austria

MA Wen-Gan

Department of Modern Physics,
University of Science and Technology of China,
Hefei, China

Franz F. SCHÖBERL

Institut für Theoretische Physik,
Universität Wien,
Boltzmanngasse 5, A-1090 Wien, Austria

*Based on invited basic lectures on numerical mathematics
(presented by Franz F. Schöberl) at the Department of Modern Physics,
University of Science and Technology of China, Hefei, China, March 1997.
Supported by the Chinese Academy of Science and
the Austrian Academic Exchange Service.*

1 Introduction

Unfortunately, most physical problems cannot be solved analytically, which means that we are not able to give their solutions in closed mathematical form. Thus, we are forced to adopt numerical methods to obtain results. But, how to keep these numerical solutions under control? In fact, one has to know—in some sense—what will emerge numerically, what will be the approximate result. Consequently, the first step, before starting the computer, will be to get a feeling for the output. First, one will analyze the dimension of the output; secondly, one will try to “guess” the order of magnitude of the desired output; and, thirdly, one will improve this first “guess.”

In these lectures we illustrate, on various examples, how to derive, even for rather complicated problems, quantities like upper bounds on energy eigenvalues analytically, and we improve our results by numerical methods. In both cases, we shall demonstrate how to use Mathematica, described as “a system for doing mathematics by computer” [1].

First, we will discuss how to handle dimensions and how to apply scaling arguments to the Schrödinger equation. After introducing the variational procedure, we calculate upper bounds on the energy eigenvalues of the Schrödinger Hamiltonian, using “paper and pencil” as well as Mathematica 3.0 [1].

One of the main advantages of the use of Mathematica is the capability to represent results very easily in graphical form. This feature is illustrated in Fig. 1 for a particular D-wave eigenfunction of the hydrogen atom.

0.8

Figure 1: The eigenfunction of the H-atom, for radial quantum number $n_r = 0$, relative orbital angular momentum $\ell = 2$, and its projection $m = 1$.

2 Dimensional Reasoning

We all know that it is a horrible task to handle dimensions. In particle physics, at least, it is customary to use a system of “natural” units defined by

$$\hbar = c = 1.$$

Using these units, we are able to express every dimensional physical quantity in, e.g., units of energy E . Clearly,

$$\text{Dim}[\hbar] = \text{Dim}[c] = 1 = \text{Dim}[E^0].$$

Then

$$\text{Dim}[Et] = \text{Dim}[E] \text{Dim}[t] = \text{Dim}[\hbar] = 1$$

and, consequently, the dimension of time t is the inverse of the dimension of energy:

$$\text{Dim}[t] = \frac{1}{\text{Dim}[E]} = \text{Dim}[E^{-1}].$$

From the well-known relation between the mass m and the rest energy E_0 of a particle,

$$E_0 = m c^2,$$

the dimension of any mass m is

$$\boxed{\text{Dim}[m] = \text{Dim}[E]} .$$

Because of

$$\text{Dim}[p] = \text{Dim}[m c] = \text{Dim}[m] ,$$

the dimension of any momentum p is also

$$\boxed{\text{Dim}[p] = \text{Dim}[E]} .$$

The dimension of any spatial coordinate x is obtained from

$$\text{Dim}[p x] = \text{Dim}[\hbar] = 1$$

as

$$\boxed{\text{Dim}[x] = \frac{1}{\text{Dim}[p]} = \text{Dim}[E^{-1}]} .$$

We are now able to express any physical quantity in units of energy. For instance, let us calculate the dimension of the Schrödinger wave function Ψ . The Schrödinger wave function $\Psi(\mathbf{x})$ in configuration-space representation is usually normalized according to

$$\int d^3x \Psi^*(\mathbf{x}) \Psi(\mathbf{x}) = 1 ,$$

which entails for the dimensions

$$\text{Dim}[x^3 \Psi^2] = 1 ,$$

hence,

$$\text{Dim}[\Psi^2] = \frac{1}{\text{Dim}[x^3]} ,$$

implying

$$\boxed{\text{Dim}[\Psi] = \text{Dim}[E^{3/2}]} .$$

Let us apply, just as an example, this method of counting dimensions to the evaluation of the Fourier transform of $1/\mathbf{p}^2$. As we know, the detailed calculation needs some care, and, involving a distribution, is not completely trivial. The quantity to be evaluated is

$$\int d^3p \frac{e^{-i\mathbf{p}\cdot\mathbf{x}}}{\mathbf{p}^2} .$$

Obviously, the only free parameter in the above integrand is $\mathbf{x}$, and since the integrand contains the rotationally invariant scalar product $\mathbf{p}\cdot\mathbf{x}$, the integral must be a function of merely the modulus $|\mathbf{x}|$ of the (spatial) coordinates $\mathbf{x}$. Now, let us count dimensions:

$$\text{Dim} \left[\frac{p^3}{p^2} \right] = \text{Dim}[p] = \text{Dim}[E] = \frac{1}{\text{Dim}[x]} .$$

Therefore, the integral has to be proportional to the inverse of $|\mathbf{x}|$:

$$\int d^3p \frac{e^{-i\mathbf{p}\cdot\mathbf{x}}}{\mathbf{p}^2} = \frac{A}{|\mathbf{x}|} .$$

The constant of proportionality A is now easily determined by applying the Laplacian $\Delta \equiv \nabla \cdot \nabla$ to both sides of this equation, and by using the rather well-known relations

$$\Delta \frac{1}{|\mathbf{x}|} = -4\pi \delta^{(3)}(\mathbf{x})$$

as well as

$$\int d^3p e^{-i\mathbf{p}\cdot\mathbf{x}} = (2\pi)^3 \delta^{(3)}(\mathbf{x}) .$$

This procedure results in

$$\Delta \int d^3p \frac{e^{-i\mathbf{p}\cdot\mathbf{x}}}{\mathbf{p}^2} = - \int d^3p e^{-i\mathbf{p}\cdot\mathbf{x}} = -(2\pi)^3 \delta^{(3)}(\mathbf{x}) = \Delta \frac{A}{|\mathbf{x}|} = -4\pi A \delta^{(3)}(\mathbf{x}) .$$

From this relation we read off for the constant of proportionality A in the above ansatz:

$$A = \frac{(2\pi)^3}{4\pi} = 2\pi^2 .$$

Just by counting dimensions and by using general physical arguments, we find, without any difficult integrations in the complex plane,

$$\boxed{\int d^3p \frac{e^{-i\mathbf{p}\cdot\mathbf{x}}}{\mathbf{p}^2} = \frac{2\pi^2}{|\mathbf{x}|}} .$$

Along the lines sketched above, we will show how to solve rather complicated problems, just by applying general theorems and rules.

3 The Schrödinger Equation

Solving the Schrödinger equation is, in general, a somewhat tedious task. However, as long as we are merely interested in the dependence of the energy eigenvalues E on the parameters entering in the Schrödinger equation, we do not have to solve the equation at all. We can adopt physical arguments only, arguments like the scaling behaviour of the Schrödinger equation.

3.1 Scaling Behaviour

The time-independent Schrödinger equation for a generic power-law central potential, i.e., a potential of the form $V(r) = a r^n$, which depends merely on the radial coordinate $r \equiv |\mathbf{x}|$, reads in configuration space, abbreviating the Laplacian again by $\Delta \equiv \nabla \cdot \nabla$,

$$\boxed{\left(-\frac{\Delta}{2\mu} + a r^n\right) \Psi(\mathbf{x}) = E \Psi(\mathbf{x})} , \tag{1}$$

with the reduced mass μ of the two-particle bound system under consideration defined by

$$\mu \equiv \frac{m_1 m_2}{m_1 + m_2} .$$

Let us start by scaling, just for fun, the coordinates $\mathbf{x}$ by some (arbitrarily chosen) scale parameter κ ,

$$\mathbf{x} = \kappa \boldsymbol{\rho} , \quad (2)$$

which gives, because of

$$\Delta = \frac{\Delta_\rho}{\kappa^2}$$

and

$$r^n = \kappa^n \rho^n ,$$

the scaled (time-independent) Schrödinger equation

$$\left(-\frac{\Delta_\rho}{2\mu\kappa^2} + a\kappa^n \rho^n \right) \Psi(\kappa \boldsymbol{\rho}) = E \Psi(\kappa \boldsymbol{\rho}) .$$

Multiplying now the above equation by $2\mu\kappa^2$ leads to

$$\left(-\Delta_\rho + 2\mu a \kappa^{2+n} \rho^n \right) \Psi = 2\mu\kappa^2 E \Psi .$$

Apart from our pure amusement, what might be the deeper reason for this re-scaling? Obviously, we can choose κ in such a way that the factor $2\mu a$ disappears:

$$\kappa^{2+n} = \frac{1}{2\mu a} ,$$

and thus

$$\kappa = \left(\frac{1}{2\mu a} \right)^{1/(2+n)} . \quad (3)$$

For this particular choice of κ , the scaled Schrödinger equation is cast into the form

$$\boxed{(-\Delta_\rho + \rho^n) \Psi = \varepsilon \Psi} , \quad (4)$$

with ε some (dimensionless) number which has to be determined by solving the scaled Schrödinger equation. Note that this scaled Schrödinger equation is dimensionless. We have thus managed to separate physics from “pure” mathematics by the identification

$$2\mu\kappa^2 E \equiv \varepsilon .$$

Inserting Eq. (3) shows us how the energy eigenvalue E depends on the parameters a , μ , and n entering in the Schrödinger equation:

$$\boxed{E = \left[\frac{a^2}{(2\mu)^n} \right]^{1/(2+n)} \varepsilon} . \quad (5)$$

Now we are in the position to discuss the physical behaviour of the energy eigenvalues E for various power-law potentials without solving explicitly the Schrödinger equation.

Coulomb potential ($n = -1$):

$$E = 2 \mu a^2 \varepsilon .$$

This energy eigenvalue is proportional to the mass and to the coupling constant squared. The energy eigenvalue E has to be proportional to the mass because the only parameter with dimension of energy entering in the Schrödinger equation is the mass μ . However, we cannot conclude from dimensional considerations only how the energy eigenvalue E depends on the (fine-structure) coupling constant a .

Linear potential ($n = 1$):

$$E = \left(\frac{a^2}{2\mu} \right)^{1/3} \varepsilon . \quad (6)$$

In contrast to the Coulomb potential, this energy eigenvalue E is proportional to the inverse of the mass μ to the power $1/3$.

Logarithmic potential ($n = 0$): Because of the (easily to check) identity

$$\ln r = \lim_{n \rightarrow 0} \frac{r^n - 1}{n} ,$$

the logarithmic-potential energy eigenvalue E is obtained for $n = 0$ in Eq. (5):

$$E = a \varepsilon .$$

Note that in this case the energy eigenvalue E is independent of the mass. That means, that, for instance, the difference of energy eigenvalues of excited states is the same even for different masses μ .

We have learned how to find the parameter dependence of the energy eigenvalues E of the Schrödinger equation without solving this differential equation explicitly. Thus we are able to check numerical solutions by calculating ratios which are independent of ε , like for, e.g., the linear potential,

$$\frac{E_1}{E_2} = \left[\left(\frac{a_1}{a_2} \right)^2 \frac{\mu_2}{\mu_1} \right]^{1/3} .$$

In this way, we are able to get a feeling for the accuracy of any numerical calculation.

3.2 The Variational Method: A Derivation of Upper Bounds

Since we are interested merely in an explicit application of the variational method, we omit all mathematically subtle discussions and explanations here. A discussion of the history of inequalities and variational methods for eigenvalue problems can be found in Ref. [2], and some applications are given in Ref. [3]. In Ref. [4], a rigorous discussion of the problem can be found. For our purpose, in the frame of these lecture notes, it is sufficient to know that, for any given Hamiltonian H with eigenvalues E_k , $k = 1, 2, \dots$, upper bounds $\hat{E}_k$ on E_k may be obtained by computing the matrix elements of H with respect to some suitably chosen set of orthogonalized trial functions Ψ_k which depend on some variational parameter λ . These upper bounds can be improved by minimizing

the resulting upper bounds $\hat{E}_k(\lambda)$ on the “true” energy eigenvalues with respect to the variational parameter λ . In order to obtain also upper bounds for the radially excited states, one has to diagonalize the corresponding energy matrix E_{ij} . The detailed steps of this general procedure are:

1. Choose a set of mutually orthogonal trial states $|\Psi_i(\lambda)\rangle$, i.e., $\langle\Psi_i(\lambda)|\Psi_j(\lambda)\rangle = \delta_{ij}$.
2. Determine the matrix elements of the Hamiltonian H using these trial functions:

$$E_{ij}(\lambda) \equiv \langle\Psi_i(\lambda)|H|\Psi_j(\lambda)\rangle .$$

3. Determine the roots of the characteristic equation

$$\det [E_{ij}(\lambda) - \hat{E}(\lambda) \delta_{ij}] = 0 . \quad (7)$$

4. These roots $\hat{E}(\lambda)$ are upper bounds for any variational parameter λ you choose.
5. You may want to improve the upper bounds by minimizing $\hat{E}(\lambda)$ with respect to λ :

$$\frac{\partial \hat{E}(\lambda)}{\partial \lambda} = 0 \quad \rightsquigarrow \quad \lambda_{\min} . \quad (8)$$

6. In this way, you obtain the minimal upper bound $\hat{E}_k(\lambda_{\min})$ on E_k (in your chosen sector of Hilbert space):

$$E_k \leq \hat{E}_k(\lambda_{\min}) . \quad (9)$$

We are now in a position to apply the above concepts to various power-law potentials.

3.2.1 Ground States

Now, let us start by considering ground states. The prototype for all studies of this kind is the Coulomb problem. The Hamiltonian H under consideration here is simply given by

$$H = \frac{\mathbf{p}^2}{2\mu} - \frac{\alpha}{r} , \quad (10)$$

with the electromagnetic fine structure constant α . First, we have to choose some trial function. Let us take, for “didactic” reasons, the hydrogen ground-state eigenfunction (of course, you can also choose, e.g., Gaussian trial functions)

$$\Psi(r, \lambda) = N e^{-\lambda r} , \quad \lambda^* = \lambda > 0 ,$$

with the variational parameter, λ , and the normalization factor, N , to be determined. Secondly, calculate the normalization factor N :

$$\int d^3x \Psi^* \Psi = 1 = N^2 \int d^3x e^{-2\lambda r} = N^2 4\pi \int_0^\infty dr r^2 e^{-2\lambda r} = N^2 4\pi \frac{\Gamma(3)}{(2\lambda)^3} ,$$

where we have used

$$\boxed{\int_0^\infty dr r^n e^{-\lambda r} = \frac{\Gamma(n+1)}{\lambda^{n+1}}} . \quad (11)$$

In Mathematica [1], the procedure is the following:

MATHEMATICA

Integration:

```
In[1]:= Integrate[r^n Exp[-lambda r],{r,0,Infinity}]
```

```
Out[1]= If[Re[lambda] > 0 && Re[n] > -1, lambda^(-1-n) Gamma[1+n],
  Integrate[r^n Exp[-lambda r],{r,0,Infinity}]]
```

This output has to be interpreted in the following way: If $\Re \lambda > 0$ and $\Re n > -1$, then the result of this integration reads $\lambda^{-1-n} \Gamma(1+n)$, otherwise, the input is returned as

$$\int_0^\infty \frac{r^n}{e^{\lambda r}} dr ,$$

which means that Mathematica is not able to find a solution. The normalization factor is thus

$$N = \frac{\lambda^{3/2}}{\sqrt{\pi}} ,$$

and the normalized trial function reads

$$\Psi(r, \lambda) = \frac{\lambda^{3/2}}{\sqrt{\pi}} e^{-\lambda r} .$$

In order to be on the safe side, we check the dimensions of this expression on both sides of this relation. As we have already discussed before, the left-hand side has dimension

$$\text{Dim}[E^{3/2}] .$$

For the exponential $\exp(-\lambda r)$ to make sense at all, λr has to be dimensionless, which requires

$$\text{Dim}[\lambda r] = 1 ,$$

leading to

$$\text{Dim}[\lambda] = \frac{1}{\text{Dim}[r]} = \text{Dim}[E] ,$$

and, therefore,

$$\text{Dim}[\Psi] = \text{Dim}[E^{3/2}] = \text{Dim}[\lambda^{3/2}] .$$

Obviously, the dimensions, at least, are correct. Now we will show how the above steps are defined and calculated using Mathematica. We give the printout as it is returned.

MATHEMATICA

Defining the trial function:

```
In[2]:= psi[r_,lambda_] := Exp[-lambda r]
```

Integration:

```
In[3]:= 4 Pi Integrate[r^2 psi[r,lambda]^2,{r,0,Infinity}]
```

```
Out[3]= 4π If[Re[lambda] > 0,  $\frac{1}{4 \lambda^3}$ ,  $\int_0^\infty e^{-2 \lambda r} r^2 dr$ ]
```

This output has to be read as: If $\Re \lambda > 0$, then the result is

$$\frac{4\pi}{4 \lambda^3} ,$$

otherwise, Mathematica returns the input,

$$4\pi \int_0^\infty e^{-2 \lambda r} r^2 dr ,$$

indicating that Mathematica is not able to evaluate this input. The complete result is

$$N^2 \frac{4\pi}{4 \lambda^3} = 1 \quad \leadsto \quad N = \sqrt{\frac{\lambda^3}{\pi}} .$$

Next, we have to evaluate the expectation values of the kinetic energy with respect to our trial function. Since we are only interested in the energy of the ground state, which is always characterized by $\ell = 0$, for our purpose the Laplacian Δ is effectively given by

$$\Delta = \frac{d^2}{dr^2} + \frac{2}{r} \frac{d}{dr} .$$

Its expectation value thus reads

$$\begin{aligned} \int d^3x \Psi^*(r, \lambda) \Delta \Psi(r, \lambda) &= \frac{\lambda^3}{\pi} \int d^3x e^{-\lambda r} \left(\frac{d^2}{dr^2} + \frac{2}{r} \frac{d}{dr} \right) e^{-\lambda r} \\ &= \frac{\lambda^3}{\pi} 4\pi \int_0^\infty dr \left(r^2 \lambda^2 - 2 r \lambda \right) e^{-2 \lambda r} \\ &= 4 \lambda^3 \left[\lambda^2 \frac{\Gamma(3)}{(2 \lambda)^3} - 2 \lambda \frac{\Gamma(2)}{(2 \lambda)^2} \right] \\ &= 4 \lambda^3 \left(-\frac{1}{4 \lambda} \right) \\ &= -\lambda^2 . \end{aligned}$$

Again we will check the dimensions and we will omit from now on the symbol $\text{Dim}[\dots]$:

$$x^3 \Psi \frac{1}{x^2} \Psi \sim E^{-3} E^{3/2} E^2 E^{3/2} = E^2 \sim \lambda^2 .$$

Fortunately, all dimensions are O.K. The expectation value of the kinetic energy is thus

$$\boxed{\left\langle \frac{\mathbf{p}^2}{2 \mu} \right\rangle = \int d^3x \Psi^*(r, \lambda) \left(-\frac{\Delta}{2 \mu} \right) \Psi(r, \lambda) = \frac{\lambda^2}{2 \mu}} . \quad (12)$$

Already knowing how to interpret Mathematica output, below we just give the output.

MATHEMATICA

Defining the trial function:

```
In[4] := psi[r_,lambda_] := lambda^(3/2)/Sqrt[Pi] Exp[-lambda r]
```

Effective Laplacian for vanishing orbital angular momentum:

```
In[5] := g[r_,lambda_] := D[psi[r,lambda],{r,2}]+2/r
D[psi[r,lambda],{r,1}]
```

Integration:

```
In[6] := 4 Pi Integrate[r^2 psi[r,lambda] g[r,lambda],{r,0,Infinity}]
```

```
Out[6]= 4π (− $\frac{\text{lambda}^2}{4\pi}$ )
```

The expression $D[\text{psi}[r,\text{lambda}],\{r,n\}]$ is the n th derivative of $\Psi(r,\lambda)$ with respect to r . The expectation value of a power-law potential $V(r) = a r^n$ reads

$$\begin{aligned} \int d^3x \Psi^*(r,\lambda) V(r) \Psi(r,\lambda) &= \frac{\lambda^3}{\pi} 4\pi a \int_0^\infty dr r^{n+2} e^{-2\lambda r} \\ &= \frac{\lambda^3}{\pi} 4\pi a \frac{\Gamma(n+3)}{(2\lambda)^{n+3}}, \end{aligned}$$

that is,

$$\boxed{\langle V(r) \rangle \equiv \int d^3x \Psi^*(r,\lambda) V(r) \Psi(r,\lambda) = 4 a \lambda^3 \frac{\Gamma(n+3)}{(2\lambda)^{n+3}}}. \quad (13)$$

For checking again the dimensions, we need the dimension of the coupling constant a :

$$V = a r^n \sim a E^{-n} \sim E$$

implies

$$a \sim E^{n+1}.$$

Without surprise, we realize that the dimensions in Eq. (13) are O.K.:

$$a \lambda^{-n} \sim E^{n+1} E^{-n} = E.$$

MATHEMATICA

Integration:

In[7] := Integrate[r^(n+2) Exp[-2 lambda r], {r, 0, Infinity}]

Out[7] = 2⁻³⁻ⁿ lambda⁻³⁻ⁿ Gamma[3+n]

Combining Eqs. (12) and (13) yields the variational energy

$$E(\lambda) = \frac{\lambda^2}{2\mu} + \frac{a}{2} \frac{\Gamma(n+3)}{(2\lambda)^n} . \quad (14)$$

This energy is always an upper bound to the “true” energy for any $\lambda > 0$: $E_{\text{true}} \leq E(\lambda)$. We are easily able to improve this upper bound $E(\lambda)$ by determining that value $\lambda_{\min}$ of the variational parameter λ which minimizes $E(\lambda)$; clearly, $\lambda_{\min}$ is then obtained from

$$\frac{\partial E(\lambda)}{\partial \lambda} = 0 .$$

The derivative of Eq. (14) with respect to λ entails the requirement

$$\frac{\partial E(\lambda)}{\partial \lambda} = \frac{\lambda}{\mu} - a n \frac{\Gamma(n+3)}{(2\lambda)^{n+1}} = 0 ,$$

which is solved by

$$\lambda_{\min} = \left[\frac{a n \mu \Gamma(n+3)}{2^{n+1}} \right]^{1/(n+2)} , \quad (15)$$

which value, when re-inserted into Eq. (14), entails, in turn, the improved upper bound

$$E_{\text{var}} \equiv E(\lambda_{\min}) = \frac{1}{2} \left(\frac{1}{\mu} \right)^{n/(n+2)} \left[\frac{a n \Gamma(n+3)}{2^{n+1}} \right]^{2/(n+2)} \left(1 + \frac{2}{n} \right) . \quad (16)$$

MATHEMATICA

Defining the function $E(\lambda)$:

In[8] := e[lambda_] := lambda^2/(2 mu) + a/2 Gamma[n+3]/(2 lambda)^n

Differentiation with respect to the parameter λ (note that $D[e[\text{lambda}], \{\text{lambda}, 1\}]$ is equivalent to $D[e[\text{lambda}], \text{lambda}]$):

In[9] := D[e[lambda], lambda]

Out[9] = $\frac{\text{lambda}}{\mu} - 2^{-1-n} a \text{lambda}^{-1-n} n \text{Gamma}[3+n]$

Solving the equation for $\lambda_{\min}$:

```
In[10]:= Solve[ $\frac{\lambda}{\mu} - 2^{-1-n} a \lambda^{-1-n} n \Gamma[3+n] == 0, \lambda]$ 
```

```
Out[10]=  $\left\{ \lambda \rightarrow \left( 2^{-1-n} a \mu n \Gamma[3+n] \right)^{\frac{1}{2+n}} \right\}$ 
```

Calculation of $E(\lambda_{\min})$:

```
In[11]:= e[( $2^{-1-n} a \mu n \Gamma[3+n]$ ) $^{\frac{1}{2+n}}$ ]
```

```
Out[11]=  $\frac{(2^{-1-n} a \mu n \Gamma[3+n])^{\frac{2}{2+n}}}{2 \mu} + 2^{-1-n} a$   

 $\Gamma[3+n] \left( (2^{-1-n} a \mu n \Gamma[3+n])^{\frac{1}{2+n}} \right)^{-n}$ 
```

Note: The command `PowerExpand[expr]` expands all powers of products and powers. % stands for the last expression (here `Out[6]`).

```
In[12]:= PowerExpand[%]
```

```
Out[12]=  $2^{\frac{-4-3n}{2+n}} a^{\frac{2}{2+n}} \mu^{-\frac{n}{2+n}} n^{-\frac{n}{2+n}} (2+n) \Gamma[3+n]^{\frac{2}{2+n}}$ 
```

`Out[12]` is the same result as in Eq. (16). Let us apply Eq. (16) to typical potentials:

Coulomb potential: Inserting here $a = -\alpha$ and $n = -1$, we obtain the upper bound

$$E_{\text{true}} \leq -\frac{\alpha^2 \mu}{2}.$$

Interestingly, the expression on the right-hand side is precisely the ground-state energy of a hydrogen atom, in other words, here the sign of equality applies. The obvious reason for this clearly is that we have used the hydrogen wave function of the ground state as our trial function. For $\alpha = 1$ and $\mu = 1$, the numerical value for E is $E = -0.5$. It is very simple to find the minimum of Eq. (14) numerically with the help of Mathematica.

MATHEMATICA

Defining the function to be minimized:

```
In[13]:= e[lambda_,n_,a_,mu_] := lambda^2/(2 mu) + a/2 Gamma[n+3]/(2  
lambda)^n
```

Finding the minimum (the starting point for the variational parameter λ is $\lambda = 0.5$):

```
In[14]:= FindMinimum[e[lambda,-1,-1,1],{lambda,0.5}]
```

```
Out[14]= {-0.5,{lambda->1.}}
```

This has to be understood as $E_{\min} = -0.5$ at the point $\lambda_{\min} = 1$.

Linear potential: $V(r) = ar$. Inserting $n = 1$ into Eq. (16) entails

$$E_{\text{true}} \leq E_{\text{var}} = \left(\frac{3}{2}\right)^{5/3} \left(\frac{a^2}{\mu}\right)^{1/3}.$$

Let us rewrite this result in the form of Eq. (6):

$$E_{\text{true}} \leq \frac{3^{5/3}}{2^{4/3}} \left(\frac{a^2}{2\mu}\right)^{1/3} = 2.4764 \left(\frac{a^2}{2\mu}\right)^{1/3}. \quad (17)$$

The true ground-state energy E_{true} is given by the first zero of the Airy function [5, 6],

$$E_{\text{true}} = 2.3381 \left(\frac{a^2}{2\mu}\right)^{1/3},$$

which implies a surprisingly small relative error of our crude upper bound E_{var} :

$$\frac{E_{\text{var}} - E_{\text{true}}}{E_{\text{true}}} \cong 6 \text{ \%}.$$

We have calculated, without solving the Schrödinger equation, the ground-state energy for the linear potential to unexpectedly good approximation analytically.

MATHEMATICA

Defining the function to be minimized:

```
In[15]:= e[lambda_,n_,a_,mu_] := lambda^2/(2 mu) + a/2 Gamma[n+3]/(2
lambda)^n
```

Finding the minimum (the starting point for the variational parameter λ is $\lambda = 0.5$):

```
In[16]:= FindMinimum[e[lambda,1,1,1],{lambda,0.5}]
```

```
Out[16]= {1.96556,{lambda->1.14471}}
```

These results may be verified by inserting $n = a = \mu = 1$ into Eqs. (17) and (15). Sometimes it is illustrative to compare plots. Letting $2\mu = 1$, we shall plot, first, $E_{\text{true}} = 2.3381 a^{2/3}$, then, $E_{\text{var}} = 2.4764 a^{2/3}$, and, finally, we shall combine these two plots into one single plot with the help of the Mathematica command **Show**.

MATHEMATICA

Defining the functions to be plotted:


```
In[17]:= etrue[a_] := 2.3381 a^(2/3)
```

```
In[18]:= eupper[a_] := 2.4764 a^(2/3)
```

Plotting E_{true} (plot1):

```
In[19]:= plot1 = Plot[etrue[a],{a,0,3},AxesLabel->{"a","E"},
TextStyle->{FontSlant->"Italic",FontSize->14}]
```

`AxesLabel` defines what shall be written on the axes; `TextStyle` defines the font type (here italic) and the font size (here 14 pt).




```
\begin{figure}[h]
\begin{center}
\psfig{figure=etrue.ps, scale=1}
\caption{etrue}
\end{center}
\end{figure}
```

The command `scale` enables you to resize the picture. For instance, `scale=0.5` would make this figure precisely half as large as the original one. The corresponding style file is included by the $\text{\TeX}$ command `\documentstyle[epsfig,...]`. In order to view the graphics, the `dvi` file has to be converted into a `ps` file.

Plotting E_{var} (plot2):

```
In[20] := plot2 = Plot[eupper[a], {a, 0, 3}, AxesLabel -> {"a", "E"},
  TextStyle -> {FontSlant -> "Italic", FontSize -> 14}]
```



Up to now we have discussed ground-state energies only. Next, we would like to show how to determine upper bounds on energies of radially excited states. A more detailed explanation of this procedure may be found in, e.g., Ref. [3]. We illustrate this method by determining the ground-state energy and the energy of the first radial excitation of a system bound by some linear potential. First, we have to choose some trial functions. For the ground state, we take again the hydrogen eigenfunction

$$\Psi_0(\lambda, r) = \frac{\lambda^{3/2}}{\sqrt{\pi}} e^{-\lambda r}, \quad \lambda^* = \lambda > 0,$$

and, for the first radially excited state, we use

$$\Psi_1(\lambda, r) = \frac{\lambda^{3/2}}{\sqrt{3\pi}} (3 - 2\lambda r) e^{-\lambda r}, \quad \lambda^* = \lambda > 0. \quad (18)$$

Since the radially excited states are states with nodes in the wave function, we have to construct trial functions with this feature. The first radially excited state has one node,

thus the trial function has to have one zero, as is the case in Eq. (18). This trial function of the first radially excited state is constructed to be orthogonal to the eigenfunction of the ground state:

$$\langle \Psi_0 | \Psi_1 \rangle = 0 .$$

To begin with, it's straightforward to check the orthogonality employing Mathematica.

MATHEMATICA

Defining the ground-state trial function Ψ_0 :

```
In[22] := psi0[lambda_,r_] := Sqrt[lambda^3/Pi] Exp[-lambda r]
```

Defining the trial function Ψ_1 for the first radially excited state:

```
In[23] := psi1[lambda_,r_] := Sqrt[lambda^3/(3 Pi)] (3-2 lambda r)
Exp[-lambda r]
```

Checking the normalization of Ψ_0 :

```
In[24] := 4 Pi Integrate[r^2 psi0[lambda,r]^2,{r,0,Infinity}]
```

```
Out[24] := 4 Pi 1/4 Pi
```

Checking the normalization of Ψ_1 :

```
In[25] := 4 Pi Integrate[r^2 psi1[lambda,r]^2,{r,0,Infinity}]
```

```
Out[25] := 4 Pi 1/4 Pi
```

Checking the orthogonality of the wave functions Ψ_0 and Ψ_1 :

```
In[26] := 4 Pi Integrate[r^2 psi0[lambda,r]
psi1[lambda,r],{r,0,Infinity}]
```

```
Out[26] = 4 Pi 0
```

The expectation value $V_{00} = \langle \Psi_0 | a r | \Psi_0 \rangle$ of the linear potential:

```
In[27]:= 4 Pi Integrate[r^2 psi0[lambda,r] a r
psi0[lambda,r],{r,0,Infinity}]
```

```
Out[27]= 4 Pi  $\frac{3 a}{8 \lambda \text{Pi}}$ 
```

The matrix elements $V_{01} = V_{10} = \langle \Psi_0 | a r | \Psi_1 \rangle$ of the linear potential:

```
In[28]:= 4 Pi Integrate[r^2 psi0[lambda,r] a r
psi1[lambda,r],{r,0,Infinity}]
```

```
Out[28]= 4 Pi  $(-\frac{\text{Sqrt}[3] a}{8 \lambda \text{Pi}})$ 
```

The expectation value $V_{11} = \langle \Psi_1 | a r | \Psi_1 \rangle$ of the linear potential:

```
In[29]:= 4 Pi Integrate[r^2 psi1[lambda,r] a r
psi1[lambda,r],{r,0,Infinity}]
```

```
Out[29]= 4 Pi  $\frac{5 a}{8 \lambda \text{Pi}}$ 
```

Defining the Laplacian of the ground-state trial function Ψ_0 :

```
In[30]:= laplacepsi0[lambda_,r_] := D[psi0[lambda,r],{r,2}]+2/r
D[psi0[lambda,r],r]
```

Defining the Laplacian of the trial function Ψ_1 for the first radially excited state:

```
In[31]:= laplacepsi1[lambda_,r_] := D[psi1[lambda,r],{r,2}]+2/r
D[psi1[lambda,r],r]
```

The expectation value $T_{00} = \left\langle \Psi_0 \left| \frac{\mathbf{p}^2}{2\mu} \right| \Psi_0 \right\rangle$ of the kinetic energy:

```
In[32]:= 4 Pi Integrate[r^2 psi0[lambda,r] (-laplacepsi0[lambda,r]/(2
mu)),{r,0,Infinity}]
```

```
Out[32]= 4 Pi  $\frac{\lambda^2}{8 \mu \text{Pi}}$ 
```

The matrix elements $T_{01} = T_{10} = \left\langle \Psi_0 \left| \frac{\mathbf{p}^2}{2\mu} \right| \Psi_1 \right\rangle$ of the kinetic energy:

```
In[33]:= 4 Pi Integrate[r^2 psi0[lambda,r] (-laplacepsi1[lambda,r]/(2
mu)),{r,0,Infinity}]
```

```
Out[33]= 4 Pi  $\frac{\lambda^2}{4 \text{Sqrt}[3] \mu \text{Pi}}$ 
```

The expectation value $T_{11} = \left\langle \Psi_1 \left| \frac{\mathbf{p}^2}{2\mu} \right| \Psi_1 \right\rangle$ of the kinetic energy:

```
In[34]:= 4 Pi Integrate[r^2 psi1[lambda,r] (-laplacepsi1[lambda,r]/(2 mu)),{r,0,Infinity}]
```

```
Out[34]= 4 Pi  $\frac{7 \lambda^2}{24 \mu}$ 
```

With the matrix elements calculated above, we are able to solve Eq. (7). However, it is much simpler to determine the energy eigenvalues using Mathematica. We shall show this analytically as well as numerically.

MATHEMATICA

The energy matrix elements:

```
In[35]:= e00[lambda_,mu_,a_] :=  $\frac{\lambda^2}{2 \mu} + \frac{3 a}{2 \lambda}$ 
```

```
In[36]:= e11[lambda_,mu_,a_] :=  $\frac{7 \lambda^2}{6 \mu} + \frac{5 a}{2 \lambda}$ 
```

```
In[37]:= e10[lambda_,mu_,a_] :=  $\frac{\lambda^2}{\sqrt{3} \mu} - \frac{\sqrt{3} a}{2 \lambda}$ 
```

Defining the energy matrix $E_{ij}(\lambda)$ as a function of λ , μ , and a :

```
In[38]:= ematrix[lambda_,mu_,a_] :=  
{ {e00[lambda,mu,a],e10[lambda,mu,a]},  
  {e10[lambda,mu,a],e11[lambda,mu,a]} }
```

Defining the energy eigenvalues $E(\lambda)$ as a function of λ , μ , and a :

```
In[39]:= eeigen[lambda_,mu_,a_] := Eigenvalues[ematrix[lambda,mu,a]]
```

Analytic evaluation of the eigenvalues:

```
In[40]:= eeigen[lambda,mu,a]
```

```
Out[40]= { (5*lambda^4*mu + 12*a*lambda*mu^2 -  
2*lambda*mu*Sqrt[4*lambda^6 - 6*a*lambda^3*mu +  
9*a^2*mu^2])/(6*lambda^2*mu^2), (5*lambda^4*mu + 12*a*lambda*mu^2 +  
2*lambda*mu*Sqrt[4*lambda^6 - 6*a*lambda^3*mu +  
9*a^2*mu^2])/(6*lambda^2*mu^2) }
```

Simplifying the above expression:

```
In[41]:= FullSimplify[%]
```

```
Out[41]= { (5*lambda^3 + 12*a*mu - 2*Sqrt[4*lambda^6 - 6*a*lambda^3*mu  
+ 9*a^2*mu^2])/(6*lambda*mu), (5*lambda^3 + 2*(6*a*mu +  
Sqrt[4*lambda^6 - 6*a*lambda^3*mu + 9*a^2*mu^2]))/(6*lambda*mu) }
```

Numerical evaluation of the eigenvalues E for $\lambda = 1$ GeV, $\mu = 0.5$ GeV, $a = 1$ GeV²:

```
In[42] := eeigen[1,1/2,1]
```

```
Out[42]= {(11 - Sqrt[13])/3, (11 + Sqrt[13])/3}
```

Numerical value (the Mathematica command `N(expr)` yields the numerical value of the expression *expr*):

```
In[43] := N[%]
```

```
Out[43]= {2.46482, 4.86852}
```

The result tells us: the variational upper bounds for the energy eigenvalues are given by

$$E_{\text{upper}}(1\text{S}) = 2.46482 \text{ GeV} ,$$

$$E_{\text{upper}}(2\text{S}) = 4.86852 \text{ GeV} ,$$

which have to be compared with the true values (determined by the first two zeros of the Airy function)

$$E_{\text{true}}(1\text{S}) = 2.33811 \text{ GeV} ,$$

$$E_{\text{true}}(2\text{S}) = 4.08795 \text{ GeV} .$$

For the ground state, labelled 1S in usual spectroscopic notation, the relative error of our variational upper bound E_{upper} is

$$\frac{E_{\text{upper}}(1\text{S}) - E_{\text{true}}(1\text{S})}{E_{\text{true}}(1\text{S})} = 5.4 \% ,$$

which is, obviously, slightly better than the previous relative error of 6 %. Accordingly, increasing the matrix size—here from 1×1 to 2×2 —improves our upper bound. For the first radially excited state, labelled 2S in usual spectroscopic notation, the relative error of our variational upper bound E_{upper} is

$$\frac{E_{\text{upper}}(2\text{S}) - E_{\text{true}}(2\text{S})}{E_{\text{true}}(2\text{S})} = 19.1 \% ,$$

which is, admittedly, a little bit large. However, we are able to improve these results by applying Eqs. (8) and (9). In this way, we find the minimum of the energy eigenvalues covered by our chosen set of trial functions.

However, varying the variational parameter λ usually destroys the orthogonality of trial functions attributed to different radial excitations. The characteristic equation is, in general, not given by Eq. (7). (A detailed discussion of these troubles can be found in Ref. [3].) In view of this, we focus our attention in the following to the ground state 1S, where this problem does not show up.

Applying the minimization procedure described at length at the beginning of this section, we try to optimize our previously obtained upper bounds on the ground-state energy of the linear potential by minimizing the corresponding eigenvalue $E(\lambda)$ of our energy matrix E_{ij} . This eigenvalue is extracted from the above analytical results using the Mathematica command `Part[expr,i]` which returns the *i*th part of the expression *expr*.

MATHEMATICA

Defining the ground-state eigenvalue as function of λ for $\mu = 0.5$ GeV and $a = 1$ GeV²:

```
In[44] := e00eigen[lambda_] :=  
Part[Eigenvalues[ematrix[lambda,1/2,1]],1]
```

Calculating the ground-state energy eigenvalue analytically:

```
In[45] := e00eigen[lambda]  
  
Out[45]= (6*lambda + 5*lambda^4 - lambda*Sqrt[9 - 12*lambda^3 +  
16*lambda^6])/(3*lambda^2)
```

Finding the minimum for the 1S-state energy (the starting point for λ is $\lambda = 0.5$ GeV):

```
In[46] := FindMinimum[%,{lambda,0.5}]
```

We might expect that the correct minimum may be found employing the Mathematica command `FindMinimum[e00eigen[lambda],{lambda,0.5}]`. Unfortunately, for very mysterious reasons, Mathematica returns a wrong result. Because of this, we carefully split the calculation into parts. Presumably, Mathematica is not able to simultaneously calculate eigenvalues, extract parts of a matrix, and find minima of some expression.

```
Out[46]= {2.4322, {lambda->0.665633}}
```

The new result, improved by minimization with respect to the variational parameter λ within the set of upper bounds obtained before, thus reads

$$E_{\text{var}}(1S) = 2.43220 \text{ GeV}$$

at the point

$$\lambda_{\text{min}} = 0.665633 \text{ GeV} .$$

Evidently, we succeeded in reducing the corresponding relative error significantly, viz., to

$$\frac{E_{\text{var}}(1S) - E_{\text{true}}(1S)}{E_{\text{true}}(1S)} = 3.9 \% .$$

As mentioned before, minimization with respect to the variational parameter leads, in general, to different values for this variational parameter in the ground state and in the radially excited states: $\lambda_i \neq \lambda_j$. Therefore, in general, all these states are no longer necessarily orthogonal:

$$\langle \Psi_i(\lambda_i) | \Psi_j(\lambda_j) \rangle \neq \delta_{ij} .$$

In this case, the characteristic equation for our eigenvalue problem generalizing Eq. (7) reads

$$\det \left[\langle \Psi_i(\lambda_i) | H | \Psi_j(\lambda_j) \rangle - \hat{E} \langle \Psi_i(\lambda_i) | \Psi_j(\lambda_j) \rangle \right] = 0 .$$

In the next subsection we will show how to increase the matrix size—and therefore the accuracy of the computed upper bounds on the Schrödinger energy levels—further.

3.2.3 Laguerre Bounds

As mentioned before, the crucial step in obtaining accurate upper bounds is the choice of convenient trial functions. Here we shall work in a basis which involves the so-called generalized Laguerre polynomials $L_k^{(\gamma)}$ [7], specific orthogonal polynomials, defined by the power series [5]

$$L_k^{(\gamma)}(x) = \sum_{j=0}^k (-1)^j \binom{k+\gamma}{k-j} \frac{x^j}{j!}$$

and normalized according to [5]

$$\int_0^\infty dx x^\gamma \exp(-x) L_k^{(\gamma)}(x) L_{k'}^{(\gamma)}(x) = \frac{\Gamma(\gamma + k + 1)}{k!} \delta_{kk'} .$$

Consequently, introducing now even two variational parameters, namely, one, λ , with the dimension of mass as well as a dimensionless one, β , our choice for the trial function $\psi_{k,\ell m}(\mathbf{x})$ with orbital angular momentum ℓ and projection m , reads in coordinate space

$$\psi_{k,\ell m}(\mathbf{x}) = \mathcal{N} |\mathbf{x}|^{\ell+\beta-1} \exp(-\lambda |\mathbf{x}|) L_k^{(\gamma)}(2\lambda |\mathbf{x}|) \mathcal{Y}_{\ell m}(\Omega) , \quad (19)$$

where normalizability constrains the variational parameter λ to positive values: $\lambda > 0$. Here, $\mathcal{Y}_{\ell m}(\Omega)$ label the spherical harmonics for angular momentum ℓ and projection m depending on the solid angle Ω ; by convention, they are orthonormalized according to

$$\int d\Omega \mathcal{Y}_{\ell m}^*(\Omega) \mathcal{Y}_{\ell' m'}(\Omega) = \delta_{\ell\ell'} \delta_{mm'} . \quad (20)$$

The proper orthonormalization of our ansatz (19) fixes the parameter γ necessarily to the value $\gamma = 2\ell + 2\beta$ and determines the normalization constant $\mathcal{N}$:

$$\psi_{k,\ell m}(\mathbf{x}) = \sqrt{\frac{(2\lambda)^{2\ell+2\beta+1} k!}{\Gamma(2\ell + 2\beta + k + 1)}} |\mathbf{x}|^{\ell+\beta-1} \exp(-\lambda |\mathbf{x}|) L_k^{(2\ell+2\beta)}(2\lambda |\mathbf{x}|) \mathcal{Y}_{\ell m}(\Omega)$$

satisfies the normalization condition

$$\int d^3x \psi_{k,\ell m}^*(\mathbf{x}) \psi_{k',\ell' m'}(\mathbf{x}) = \delta_{kk'} \delta_{\ell\ell'} \delta_{mm'} .$$

Rather obviously, normalizability constrains the second variational parameter, β , too, namely, to a range characterized by $2\beta > -1$, i.e., to the range $\beta > -\frac{1}{2}$. For simplicity, we set our mass scale by choosing $m_1 = m_2 = 1$ GeV and fix the variational parameters to the values $\lambda = 1$ GeV and $\beta = 1$. We illustrate the general procedure by discussing, for comparison, again the linear potential $V = r$, for simplicity with slope $a = 1$ GeV². At least up to a 4×4 matrix, all calculations may be performed analytically; we leave this to the reader as some exercise, and follow our procedure by applying Mathematica:

- define the chosen set of trial functions $\psi_{k,\ell m}(\mathbf{x})$;
- compute the matrix elements of the Laplacian, in other words, the kinetic energy;
- compute the matrix elements of the radial coordinate r , i.e., the potential energy;
- determine the eigenvalues $E(\lambda)$ of the resulting matrix $E_{ij}(\lambda)$ of the total energy;
- compare the eigenvalues for different matrix sizes in order to observe, hopefully, convergence.

MATHEMATICA

Defining the trial function $\psi_{k,\ell m}(\mathbf{x})$:

```
In[47] := psix[k_,l_,m_,r_] := Sqrt[2^(2 l+3) k!/Gamma[2 l+3+k]] r^l
Exp[-r]*LaguerreL[k,2 l+2,2 r]*SphericalHarmonicY[l,m,theta,phi]
```

Since we discuss S waves only, we define the trial function for $\ell = m = 0$:

```
In[48] := psi[k_,r_] := psix[k,0,0,r]
```

Defining the Laplacian $\Delta\psi_k(r)$ operating on states with $\ell = 0$ (S waves):

```
In[49] := delta[k_,r_] := D[psi[k,r],{r,2}]+2/r D[psi[k,r],{r,1}]
```

Defining the integrand $\psi_s(r) \Delta\psi_k(r)$:

```
In[50] := intks[k_,s_,r_] := psi[s,r] delta[k,r]
```

The matrix elements $\int_0^\infty dr r^2 \psi_s(r) (-\Delta\psi_k(r))$ of the kinetic-energy operator $T = -\Delta$:¹

```
In[51] := kinen[k_,s_] := -4 Pi Integrate[r^2
intks[k,s,r],{r,0,Infinity}]
```

The matrix elements $\int_0^\infty dr r^2 \psi_s(r) r \psi_k(r)$ of the potential-energy operator $V(r) = r$:

```
In[52] := poten[k_,s_] := 4 Pi Integrate[r^3 psi[s,r]
psi[k,r],{r,0,Infinity}]
```

The matrix elements of the total energy:

```
In[53] := toten[k_,s_] := kinen[k,s]+poten[k,s]
```

Here we construct the matrix built by the above matrix elements, using the command `Table`. Since counting of the matrix indices starts at 0, we redefine this matrix in order to obtain for $x = 1$ a 1×1 matrix, and so on.

```
In[54] := totenmat[x_] := Table[toten[k,s],{k,0,x-1},{s,0,x-1}]
```

Defining now the function `eeigen[x]` with the help of the command `Eigenvalues[M]` which gives the eigenvalues of an $x \times x$ matrix M , in other words, diagonalizes any $x \times x$ matrix (for our purposes, it has to be applied, of course, to the matrix `totenmat[x]` of the total energy):

```
In[55] := eeigen[x_] := Eigenvalues[totenmat[x]]
```

¹ Since we have chosen both particle masses to be $m_1 = m_2 = 1$ GeV, twice the reduced mass μ is also 1 GeV.

Eigenvalue of the 1×1 energy matrix:

In[56] := eeigen[1]

Out[56] = $\left\{\frac{5}{2}\right\}$

Eigenvalues of the 2×2 energy matrix:

In[57] := eeigen[2]

Out[57] = $\left\{\frac{1}{3}(11 - \sqrt{13}), \frac{1}{3}(11 + \sqrt{13})\right\}$

Using the command N[%] for numerical evaluation of the last output:

In[58] := N[%]

Ground-state energy and energy of the first radial excitation numerically:

Out[58] = {2.46482, 4.86852}

Eigenvalues of the 3×3 energy matrix:

In[59] := eeigen[3]

Out[59] = $\left\{\frac{9}{2}, 5 - \sqrt{7}, 5 + \sqrt{7}\right\}$

Using the command N[%] for numerical evaluation of the last output:

In[60] := N[%]

Ground-state energy and energies of first and second radial excitations numerically:

Out[60] = {4.5, 2.35425, 7.64575}

In general, the eigenvalues of a 5×5 matrix cannot be found analytically; we determine them numerically:

In[61] := N[eeigen[5]]

Ground-state energy and energies of the first four radial excitations:

Out[61] = {2.34136, 4.13334, 5.72535, 8.11424, 15.519}

The computation of the eigenvalues of the 10×10 energy matrix consumes already a noticeable amount of computer time:

In[62] := N[eeigen[10]]

Ground-state energy and energies of the first nine radial excitations:

Out[62] =
{2.33812, 4.08858, 5.53209, 6.83859, 8.14892, 9.91409, 12.195, 14.096, 17.146,
49.7026}

Table 1 compares the dependence of the accuracy of the obtained bounds on the energy levels of ground state and first radially excited state on the size of the energy matrix.

Table 1: Comparison of relative errors of the energy eigenvalues for varying matrix sizes

Matrix Size	1S State	2S State
1×1	6 %	—
2×2	5 %	19 %
3×3	0.7 %	10 %
5×5	0.1 %	1 %
10×10	4×10^{-4} %	2×10^{-2} %

Obviously, the accuracy obtained is very impressive. One should keep in mind that the above procedure is applicable not only to potentials $V(r)$ different from the power-law form $V = r^n$ but also to differential operators different from the nonrelativistic $\mathbf{p}^2/2m$. In this way one is able to determine even the energy eigenvalues of more sophisticated Hamiltonians like the semirelativistic one, which incorporates the square-root operator $\sqrt{\mathbf{p}^2 + m^2}$ of the relativistic kinetic energy [7]. Moreover, it might be of interest to note that up to and including 4×4 matrices, the diagonalization of the energy matrix E can be performed analytically. This fact enables one to gain control over purely numerically obtained results. However, one should never forget that the numerical values computed here represent always only upper bounds on the true energy eigenvalues: $E_{\text{true}} \leq E(\lambda)$.

In summary, we have tried to show how (rather complicated) bound-state problems can be handled “easily,” using Mathematica and a few important theorems to calculate at least the resulting energy levels to a high precision. We didn’t touch upon the task of determining also the corresponding wave functions. Not surprisingly, one can show [8] that, for matrix sizes large enough, an arbitrary accuracy for the wave functions can be achieved too. On the other hand, for small matrices the general conclusion is that one should not trust the wave functions at all, even if the energy levels are rather accurate.

For the moment, let’s only produce three-dimensional plots of our trial function for $k = 0$ and $k = 5$, to get a feeling how it looks like. The plots are scaled with `scale=0.7`.

MATHEMATICA

Reduced trial function $\psi_{k,\ell m}(\mathbf{x})$, to be plotted as a function $\psi_{k,\ell m}(x, y)$ of x and y only:

```
In[63] := psix[k_, l_, m_, x_, y_] := Sqrt[2^(2 l+3) k!/Gamma[2 l+3+k]]
Sqrt[x^2+y^2]^l Exp[-Sqrt[x^2+y^2]] LaguerreL[k, 2 l+2, 2 Sqrt[x^2+y^2]]
SphericalHarmonicY[l, m, theta, phi]
```

Plotting the trial function $\psi_{k,\ell m}(x, y)$ for $k = \ell = m = 0$ as a three-dimensional plot. This function represents by itself a suitable trial function, namely, for the ground state.

```
In[64] := Plot3D[psix[0,0,0,x,y],{x,-4,4},{y,-4,4},
TextStyle->{FontSlant->"Italic",FontSize->12}]
```

Plotting the trial function $\psi_{k,\ell m}(x, y)$ for $k = 5$ and $\ell = m = 0$ as a three-dimensional plot. For $k \neq 0$, the functions $\psi_{k,\ell m}$ are not trial functions corresponding to particular levels of excitation. Adequate trial functions have to be determined by calculating the eigenvectors of the energy matrix; they will be superpositions of various trial functions.


```
In[65] := Plot3D[psix[5,0,0,x,y],{x,-4,4},{y,-4,4},
TextStyle->{FontSlant->"Italic",FontSize->12}]
```

Acknowledgement

One of us, F. F. S., would like to thank the University of Science and Technology of China and the Institute for High Energy Physics, Beijing, of the Academy of Science of China for their hospitality during the stay in China.


References

- [1] S. Wolfram, *Das Mathematica Buch: Mathematica Version 3* (Addison-Wesley-Longman, Bonn, 1997).
- [2] A. Weinstein and W. Stenger, *Methods of Intermediate Problems for Eigenvalues – Theory and Ramifications* (Academic Press, New York, 1972).
- [3] D. Flamm and F. Schöberl, *Introduction to the Quark Model of Elementary Particles* (Gordon and Breach, New York, 1982).
- [4] M. Reed and B. Simon, *Methods of Modern Mathematical Physics IV: Analysis of Operators* (Academic Press, New York, 1978) Sections XIII.1 and XIII.2.
- [5] *Handbook of Mathematical Functions*, editors M. Abramowitz and I. A. Stegun (Dover, New York, 1964).
- [6] W. Lucha, F. F. Schöberl, and D. Gromes, Phys. Reports **200** (1991) 127.
- [7] W. Lucha and F. F. Schöberl, Phys. Rev. A **56** (1997) 139; hep-ph/9609322.
- [8] W. Lucha and F. F. Schöberl, Vienna preprint HEPHY-PUB 693/98, UWThPh-1998-38 (in preparation).



0.7

Figure 5: The trial function $\psi_{k,\ell m}(x, y)$ for $k = \ell = m = 0$



0.7

Figure 6: The trial function $\psi_{k,m}(x,y)$ for $k=5$ and $\ell=m=0$

